Modelling and Optimization

INF170

#6:Transportation and Assignment Problem

AHMAD HEMMATI

Optimization Group
Dept. of Informatics
University of Bergen

Fall Semester
2018

AGENDA

- Transportation Model
- Assignment Model

Transportation Problem

A single good is to be shipped from several origins to several destinations at minimum overall cost.

Transportation Problem

Suppose that we have decided to produce steel coils at three mill locations, in the following amounts:

GARY	Gary, Indiana	1400
CLEV	Cleveland, Ohio	2600
PITT	Pittsburgh, Pennsylvania	2900
SUM		**6900**

Transportation Problem

The total of 6,900 tons must be shipped in various amounts to meet orders at seven locations of automobile factories:

FRA	Framingham, Massachusetts	900
DET	Detroit, Michigan	1200
LAN	Lansing, Michigan	600
WIN	Windsor, Ontario	400
STL	St. Louis, Missouri	1700
FRE	Fremont, California	1100
LAF	Lafayette, Indiana	1000
SUM		6900

Transportation Problem

Table of shipping costs per ton:

	GARY	CLEV	PITT
FRA	39	27	24
DET	14	9	14
LAN	11	12	17
WIN	14	9	13
STL	16	26	28
FRE	82	95	99
LAF	8	17	20

What is the least expensive plan for shipping the coils from mills to plants?

Transportation Problem



Transportation Problem



Transportation Problem

Variables:

GARY:FRA → the number of tons to be shipped from GARY to FRA

… (21 decision variables in all! One for each combination of mill and factory)

Objective function: *Minimize*

39 GARY:FRA + 27 CLEV:FRA + 24 PITT:FRA +

14 GARY:DET + 9 CLEV:DET + 14 PITT:DET +

11 GARY:LAN + 12 CLEV:LAN + 17 PITT:LAN +

14 GARY:WIN + 9 CLEV:WIN + 13 PITT:WIN +

16 GARY:STL + 26 CLEV:STL + 28 PITT:STL +

82 GARY:FRE + 95 CLEV:FRE + 99 PITT:FRE +

8 GARY:LAF + 17 CLEV:LAF + 20 PITT:LAF

	GARY	CLEV	PITT
FRA	39	27	24
DET	14	9	14
LAN	11	12	17
WIN	14	9	13
STL	16	26	28
FRE	82	95	99
LAF	8	17	20

Transportation Problem

What about supplying the factories from the mill that can ship most cheaply to it!

	GARY	CLEV	PITT
FRA	39	27	24
DET	14	9	14
LAN	11	12	17
WIN	14	9	13
STL	16	26	28
FRE	82	95	99
LAF	8	17	20

	GARY	CLEV	PITT
FRA	0	0	900
DET	0	1200	0
LAN	600	0	0
WIN	0	400	0
STL	1700	0	0
FRE	1100	0	0
LAF	1000	0	0
SUM	~~**4400**~~	~~**1600**~~	~~**900**~~

Transportation Problem

Shipments from GARY to the seven factories is equal to the production level of 1400

GARY:FRA + GARY:DET + GARY:LAN + GARY:WIN + GARY:STL + GARY:FRE + GARY:LAF = 1400

Transportation Problem

GARY:FRA + GARY:DET + GARY:LAN + GARY:WIN + GARY:STL + GARY:FRE + GARY:LAF = 1400

CLEV:FRA + CLEV:DET + CLEV:LAN + CLEV:WIN + CLEV:STL + CLEV:FRE + CLEV:LAF = 2600

PITT:FRA + PITT:DET + PITT:LAN + PITT:WIN + PITT:STL + PITT:FRE + PITT:LAF = 2900

At FRA, the sum of the shipments received from the three mills must equal the 900 tons ordered:

$$GARY{:}FRA + CLEV{:}FRA + PITT{:}FRA = 900$$

Transportation Problem

GARY:FRA + CLEV:FRA + PITT:FRA = 900

GARY:DET + CLEV:DET + PITT:DET = 1200

GARY:LAN + CLEV:LAN + PITT:LAN = 600

GARY:WIN + CLEV:WIN + PITT:WIN = 400

GARY:STL + CLEV:STL + PITT:STL = 1700

GARY:FRE + CLEV:FRE + PITT:FRE = 1100

GARY:LAF + CLEV:LAF + PITT:LAF = 1000

Two fundamental sets of objects:

- ✓ The sources or origins (mills, in our example)
- ✓ The destinations (factories , in our example)

Declaration of these two sets in AMPL model :

```
set ORIG;
set DEST;
```

Transportation Problem

There is

- a supply of something at each origin (tons of steel coils produced, in our case)
- a demand for the same thing at each destination (tons of coils ordered).

```
param supply {ORIG} >= 0;
param demand {DEST} >= 0;
```

```
check: sum {i in ORIG} supply[i] = sum {j in DEST} demand[j];
```

Transportation Problem

For each combination of an origin and a destination, there is a transportation cost and a variable representing the amount transported.

```
param cost {ORIG,DEST} >= 0;
var Trans {ORIG,DEST} >= 0;
```

For a particular origin *i* and destination *j*,

We ship *Trans*[*i*,*j*] units from *i* to *j*, at a cost of *cost*[*i*,*j*] per unit

- The total cost for this pair is

```
cost[i,j] * Trans[i,j]
```

Adding over all pairs, we have the objective function:

```
minimize Total_Cost:
sum {i in ORIG, j in DEST} cost[i,j] * Trans[i,j];
```

These are the same:

```
sum {i in ORIG, j in DEST} cost[i,j] * Trans[i,j];
```

```
sum {j in DEST, i in ORIG} cost[i,j] * Trans[i,j];
```

```
sum {i in ORIG} sum {j in DEST} cost[i,j] * Trans[i,j];
```

Constraints:

Declaration in AMPL model:

```
subject to Supply {i in ORIG}: ...
subject to Demand {j in DEST}: ...
```

- Note that the names supply and Supply are unrelated
- AMPL distinguishes upper and lower case.

Constraints:

Supply constraint for origin *i*:

- The sum of all shipments out of *i* is equal to the supply available.

```
subject to Supply {i in ORIG}:
sum {j in DEST} Trans[i,j] = supply[i];
```

Constraints:

Demand constraint for destination *j*:

- The sum of all shipments into *j* is equal to the demand available.

```
subject to Demand {j in DEST}:
sum {i in ORIG} Trans[i,j] = demand[j];
```

Transportation Problem

```
set ORIG; # origins
set DEST; # destinations
param supply {ORIG} >= 0; # amounts available at origins
param demand {DEST} >= 0; # amounts required at destinations
check: sum {i in ORIG} supply[i] = sum {j in DEST} demand[j];
param cost {ORIG,DEST} >= 0; # shipment costs per unit
var Trans {ORIG,DEST} >= 0; # units to be shipped
minimize Total_Cost:
sum {i in ORIG, j in DEST} cost[i,j] * Trans[i,j];
subject to Supply {i in ORIG}:
sum {j in DEST} Trans[i,j] = supply[i];
subject to Demand {j in DEST}:
sum {i in ORIG} Trans[i,j] = demand[j];
```

Transportation Problem



	CLEV	GARY	PITT
DET	1200	0	0
FRA	0	0	900
FRE	0	1100	0
LAF	400	300	300
LAN	600	0	0
STL	0	0	1700
WIN	400	0	0

Transportation Problem



Transportation Problem



	CLEV	GARY	PITT
DET	1200	0	0
FRA	0	0	900
FRE	0	1100	0
LAF	400	0	600
LAN	600	0	0
STL	0	300	1400
WIN	400	0	0

Transportation Problem

$$minimize \sum_{i \in N_1} \sum_{j \in N_2} c_{i,j} x_{i,j}$$

subject to

$\sum_{j \in N_2} x_{i,j} = s_i$ for all $i \in N_1$ (Supply)

$\sum_{i \in N_1} x_{i,j} = d_j$ for all $j \in N_2$ (Demand)

$x_{i,j} \geq 0$ for all $i \in N_1$ and $j \in N_2$(Nonnegativity)

Assignment Problem

Consider a department that needs to assign some number of people to an equal number of offices.

- The origins
- Destinations

Assignment Problem

Consider a department that needs to assign some number of people to an equal number of offices.

- The origins now represent individual people
- Destinations represent individual offices

Assignment Problem

- Since each person is assigned one office, and each office is occupied by one person,
  - $supply[i] = 1$
  - $demand[j] = 1$
- Interpret $Trans[i,j]$ as the "amount" of person i that is assigned to office j
  - $Trans[i,j]$ is 1 then person i will occupy office j
  - $Trans[i,j]$ is 0 then person i will not occupy office j.

Objective function?

- ✓ Ask people to rank the offices, giving their first choice, second choice, and so forth. Then let *cost*[i,j] be the rank that person i gives to office j.
- ✓ Assignment that will please a lot of people!
- ✓ Same model as transportation model but new data!

Assignment Problem

```
set ORIG := Coullard Daskin Hazen Hopp Iravani Linetsky Mehrotra Nelson
Smilowitz Tamhane White ;
set DEST := C118 C138 C140 C246 C250 C251 D237 D239 D241 M233 M239;
param supply default 1 ;
param demand default 1 ;
param cost:
          C118 C138 C140 C246 C250 C251 D237 D239 D241 M233 M239 :=
Coullard  6    9    8    7    11   10   4    5    3    2    1
Daskin    11   8    7    6    9    10   1    5    4    2    3
Hazen     9    10   11   1    5    6    2    7    8    3    4
Hopp      11   9    8    10   6    5    1    7    4    2    3
Iravani   3    2    8    9    10   11   1    5    4    6    7
Linetsky  11   9    10   5    3    4    6    7    8    1    2
Mehrotra  6    11   10   9    8    7    1    2    5    4    3
Nelson    11   5    4    6    7    8    1    9    10   2    3
Smilowitz 11   9    10   8    6    5    7    3    4    1    2
Tamhane   5    6    9    8    4    3    7    10   11   2    1
White     11   9    8    4    6    5    3    10   7    2    1 ;
```

Assignment Problem

Coullard	C118	6
Daskin	D241	4
Hazen	C246	1
Hopp	D237	1
Iravani	C138	2
Linetsky	C250	3
Mehrotra	D239	2
Nelson	C140	4
Smilowitz	M233	1
Tamhane	C251	3
White	M239	1

Assignment Problem

$$minimize \sum_{(i,j)\in A} c_{i,j}x_{i,j}$$

subject to

$\sum_{(i,k)\in A} x_{i,k} = 1$ for all $i \in N_1$ (Supply)

$\sum_{(k,j)\in A} x_{k,j} = 1$ for all $j \in N_2$ (Demand)

$x_{i,j} \geq 0$ for all $(i, j) \in A$ (Nonnegativity)

Assignment Problem

But how did we know that every *Trans*[*i*,*j*] would equal either 0 or 1 in the optimal solution, rather than, say, 0.5?

Special property of transportation models, which guarantees that

- ✓ as long as all supply and demand values are integers, and
- ✓ as long as all all lower and upper bounds on the variables are integers,

there will be an optimal solution that is entirely integral.

But don't let this favorable result mislead you into assuming that integrality can be assured in all other circumstances; even in examples that seem to be much like the transportation model, finding integral solutions can require a special solver, and a lot more work.

Transportation Problem

$$minimize \sum_{i \in N_1} \sum_{j \in N_2} c_{i,j} x_{i,j}$$

subject to

$\sum_{j \in N_2} x_{i,j} = s_i$ for all $i \in N_1$ (Supply)

$\sum_{i \in N_1} x_{i,j} = d_j$ for all $j \in N_2$ (Demand)

$x_{i,j} \geq 0$ for all $i \in N_1$ and $j \in N_2$ (Nonnegativity)

Transportation Problem

$\boxed{\sum s_i \geq \sum d_j}$

$$minimize \sum_{i \in N_1} \sum_{j \in N_2} c_{i,j} x_{i,j}$$

subject to

$\sum_{j \in N_2} x_{i,j} \leq s_i$ for all $i \in N_1$ (Supply)

$\sum_{i \in N_1} x_{i,j} \geq d_j$ for all $j \in N_2$ (Demand)

$x_{i,j} \geq 0$ for all $i \in N_1$ and $j \in N_2$ (Nonnegativity)

Transportation Problem

$$minimize \sum_{i \in N_1} \sum_{j \in N_2} c_{i,j} x_{i,j}$$

subject to

$\sum_{j \in N_2} x_{i,j} \leq s_i$ for all $i \in N_1$ (Supply)

$\sum_{i \in N_1} x_{i,j} \leq d_j$ for all $j \in N_2$ (Demand)

$x_{i,j} \geq 0$ for all $i \in N_1$ and $j \in N_2$ (Nonnegativity)

$\boxed{x_{i,j} = 0 \rightarrow Obj = 0}$

Transportation Problem

$\boxed{\sum s_i \leq \sum d_j}$

$$minimize \sum_{i \in N_1} \sum_{j \in N_2} c_{i,j} x_{i,j}$$

subject to

$\sum_{j \in N_2} x_{i,j} \geq s_i$ for all $i \in N_1$ (Supply)

$\sum_{i \in N_1} x_{i,j} \leq d_j$ for all $j \in N_2$ (Demand)

$x_{i,j} \geq 0$ for all $i \in N_1$ and $j \in N_2$ (Nonnegativity)

Use of Dummy Sources!

- ✓ Add a dummy Origin in case of shortage in supply!
  - ➢ With capacity equal to shortage and new links with high costs
- ✓ Add a dummy destination in case of shortage in demand!
  - ➢ With capacity equal to shortage and new links with high costs

Use of Penalty!

Transportation Problem

$$minimize \sum_{i\in N_1}\sum_{j\in N_2} c_{i,j}x_{i,j} + P\left[\sum_{j\in N_2}\left(d_j - \sum_{i\in N_1} x_{i,j}\right)\right]$$

subject to

$\sum_{j\in N_2} x_{i,j} \le s_i$ for all $i \in N_1$ (Supply)

$\sum_{i\in N_1} x_{i,j} \le d_j$ for all $j \in N_2$ (Demand)

$x_{i,j} \ge 0$ for all $i \in N_1$ and $j \in N_2$ (Nonnegativity)

TRANSPORTATION PROBLEM

$$minimize \sum_{i \in N_1} \sum_{j \in N_2} c_{i,j} x_{i,j} + \sum_{j \in N_2} P_j \left(d_j - \sum_{i \in N_1} x_{i,j} \right)$$

subject to

$\sum_{j \in N_2} x_{i,j} \leq s_i$ for all $i \in N_1$ (Supply)

$\sum_{i \in N_1} x_{i,j} \leq d_j$ for all $j \in N_2$ (Demand)

$x_{i,j} \geq 0$ for all $i \in N_1$ and $j \in N_2$ (Nonnegativity)

ASSIGNMENT #3:

AMPL BOOK
CHAPTER 3. EXERCISES(1-4)

IN CLASS ASSIGNMENT (TEST) #1:

- ✓ Closed-book
- ✓ Based on slides
- ✓ No need of AMPL

Next Lecture

Lecture #7:

Network Optimization